

Pig latin queries

student.txt

```

    john,senior,cse
    mary,senior,cse
    bob,junior,ece
    lee,sophomore,ece

```

transcript.txt

```

    john,cse482,3.5
    john,cse335,3.5
    mary,cse482,3.5
    mary,cse335,4.0
    bob,cse335,2.5
    bob,cse232,3.0
    lee,cse335,3.5

```

A.

```

data = LOAD 'transcript.txt' USING PigStorage(',') AS (Sname, Course,
GPA:float);
tmp = FILTER data BY Course == 'cse482';
result = FOREACH tmp GENERATE Sname, GPA;
DUMP result;

```

Finds students and their GPA who have taken cse482

(john, 3.5)

(mary, 3.5)

B.

```

data = LOAD 'transcript.txt' USING PigStorage(',') AS (Sname, Course,
GPA:float);
grp = GROUP data BY Course;
tmp = FOREACH grp GENERATE group AS Course, AVG(data.$2) AS avgGPA;
result = ORDER tmp BY avgGPA desc;
dump result;

```

Finds average GPA for each course

(cse482,3.5)

(cse335,3.375)

(cse232,3.0)

C.

```

std = LOAD 'student.txt' USING PigStorage(',') AS (Name, Status, Dept);
cs = FILTER std BY Dept == 'cse';
data = LOAD 'transcript.txt' USING PigStorage(',')
      AS (Sname, Course, GPA:float);
tmp = JOIN cs by Name, data by Sname;
tmp2 = FOREACH tmp GENERATE Sname, Course;
result = DISTINCT tmp2;

```

dump result;

Finds unique pairs of CSE students and classes they took.

```
(john,cse335)
(john,cse482)
(mary,cse335)
(mary,cse482)
```

D.

```
std = LOAD 'student.txt' USING PigStorage(',');
data = LOAD 'transcript.txt' USING PigStorage(',');
enrollment = JOIN std by $0, data by $0;
tmp = FOREACH enrollment GENERATE $0, $2, $4, $5;
tmp2 = GROUP tmp BY $2;
result = FOREACH tmp2 GENERATE group, MIN(tmp.$3);
dump result;
```

Minimum GPA for each course.

```
(cse232,3.0)
(cse335,2.5)
(cse482,3.5)
```

E.

```
data = LOAD 'transcript.txt' USING PigStorage(',');
grp1 = GROUP data BY $1;
tmp = FOREACH grp1 GENERATE group, 'all', AVG($1.$2);
grp2 = GROUP data all;
tmp2 = FOREACH grp2 GENERATE $0, AVG($1.$2);
tmp3 = JOIN tmp BY $1, tmp2 BY $0;
tmp4 = FILTER tmp3 BY $2 > $4;
result = FOREACH tmp4 GENERATE $0;
dump result;
```

Finds course average GPA that is above the total average GPA.

```
(cse482)
(cse335)
```